PharmaSUG 2020 - Paper SS-030

End-to-end Prostate-Specific Antigen (PSA) Analysis in Clinical Trials: From Mock-ups to ADPSA

Joy Zeng, Pfizer Inc;
Varaprasad Ilapogu, Ephicacy Consultancy Group;
Xinping Cindy Wu, Pfizer Inc;

ABSTRACT

Prostate-specific antigen (PSA) level is a key biomarker in prostate cancer that has been used in standard guidelines as a measurement of clinical outcomes for patients with prostate cancer. This paper aims to provide an end-to-end overview of the programming aspects of PSA-related trials. We describe the concepts of PSA response and time to PSA progression, two important end points in assessing efficacy of prostate cancer trials, along with the statistical methods involved in estimating the distribution of time to PSA progression. The paper also addresses the design of metadata from PSA-related mock-up tables and presents the considerations involved in the creation of CDISC-compliant ADPSA dataset based on the metadata. Programming in the oncology therapeutic area is highly specialized and we hope this paper serves as a one-stop shop for providing the necessary tools to navigate through it.

INTRODUCTION

Prostate cancer is the most common cancer among men in the United States and is associated with a high risk of bone metastases and death. The American Cancer Society estimates that in 2019, up to 174,650 men in the United States were diagnosed with prostate cancer and approximately 31,620 will die of the prostate cancer disease [1]. While other diagnostic tools have been developed for prostate cancer, Prostate-specific antigen (PSA) blood test continues to be the main biomarker in monitoring the progression of prostate cancer.

This paper specifically focuses on PSA-related analysis that is typically conducted in prostate cancer trials. PSA-related analyses can be categorized into 3 types: descriptive summary of PSA changes from Baseline by visit, categorical analysis for summarizing PSA response rate, and time-to-event analysis to estimate and compare survival experiences of different groups in a randomized controlled trial. We hope to provide readers with an overview of the necessary information in statistical programming.

DATA COLLECTION

In clinical trials, PSA measurements are collected through laboratory tests and the analysis results are presented in the Laboratory Tests (LB) domain-related outputs. An example of an annotated case report form (CRF) that shows how PSA data is collected in the LB domain is presented below:



Figure 1. Sample CRF Page for PSA Data

PSA-RELATED ANALYSIS

PSA RESPONSE

Prostate Cancer Working Group 2 (PCWG2) provides a standard guideline for definition of PSA response [2]. The current version of Prostate Cancer Working Group is PCWG3, but this paper uses PCWG2 as it is the current version our internal studies utilize. The population of PSA evaluable patients is defined as those patients with a baseline PSA value and at least one postbaseline PSA value. PSA response is calculated at each visit based on the decline from baseline in PSA among all those PSA evaluable patients. If a subject has any post-baseline PSA values with a decrease from baseline >=50% (or >=30%, or >=90%), followed by a consecutive value that also has a decrease from baseline >=50% (or >=30%, or >=90%) and is conducted at least 3 weeks later, then the subject is considered to have a PSA response. If a consecutive value meets the response criteria but is obtained within 3 weeks and the next assessment also meets response criteria and is taken after 3 weeks, then the initial response is considered as confirmed response as well. However, a subject with missing PSA value is considered as non-responder.

In the analysis of best PSA response rate, only patients who have both baseline and at least one postbaseline assessments are included. Additionally, only assessments performed from the date of randomization until confirmed PSA progression or start of new anticancer treatment are considered in the analysis. When summarizing unconfirmed PSA response rate, both confirmed and unconfirmed responses should be included. In other words, as long as a subject has any post-baseline PSA value with a decrease from baseline >=50% % (or >=30%, or >=90%), it should be considered as an unconfirmed response. An example of a reporting table for PSA response rate is shown below:

Table 1.0
PSA Response Rate (Decrease from Baseline >=50%)
(Evaluable ITT Population)

	Treatment(N=XXX)	Placebo(N=XXX)	Treatment Comparison Drug vs. Placebo
Number of Patients with PSA Values at Baseline	XXX(XX.X%)	XXX(XX.X%)	
With At Least One Postbaseline PSA Assessment	XXX(XX.X%)	XXX(XX.X%)	
No Postbaseline Assessment	XXX(XX.X%)	XXX(XX.X%)	
Change in PSA from Baseline to PSA Nadir(Confirmed)	XXX	XXX	
Responders(>=50% Reduction)	XXX(XX.X%)	XXX(XX.X%)	
Non-Responders	XXX(XX.X%)	XXX(XX.X%)	
95%CI for Response Rate	XX.X%-XX.X%	XX.X%-XX.X%	
Difference in Response Rate(95% CI)			XX.XX%(XX.XX%-XX.XX%)
P-value			X.XXXX
Change in PSA from Baseline to PSA Nadir (Including both Confirmed and Unconfirmed)	XXX	XXX	
Responders(>=50% Reduction)	XXX(XX.X%)	XXX(XX.X%)	
Non-Responders	XXX(XX.X%)	XXX(XX.X%)	
95%CI for Response Rate	XX.X%-XX.X%	XX.X%-XX.X%	
Difference in Response Rate(95% CI)			XX.XX%(XX.XX%-XX.XX%)
P-value			X.XXXX

In the summary table for PSA response, the proportion of patients in PSA evaluable population with maximal PSA declines of at least 50%, along with the associated 95% confidence intervals for each arm is reported using Clopper-Pearson binomial confidence interval (CI) [3]. The difference in response rate is calculated by the response rate in treatment arm minus the rate in placebo. A two-sided interval based on an approximate normal distribution (The 'riskdiff' option in *SAS*® specifies the difference is calculated based on a standard normal approximation). Additionally, the proportion of patients with a confirmed >=50% reduction in PSA from baseline is usually compared between the treatment arm and placebo using Cochran-Mantel-Haenszel (CMH) score test.

Some useful *SAS*® function for the analysis of PSA response is attached below:

```
*************************************************************************
  95%CI for Response rate: Based on Exact Binomial test(Clopper-Pearson)
*************************************************************************;
    proc freq data=adpsa_ noprint;
          by trtpn;
          table resp50fl/binomial(exact);
          output out=resprt(keep=trtpn XL_BIN XU_BIN) Binomial;
    run;
*****************************************************************
  Difference in response: based on Standard Normal Approximation
*****************************************************************;
    proc freq data=adpsa_ noprint;
          table trtpn*resp50fl/riskdiff;
          output out=diffrt(keep=N _RDIF1_ L_RDIF1 U_RDIF1) riskdiff;
    Run;
*********************************************************
   P-value: based on unstratified CMH mean score test
*********************************************************;
    proc freq data=adpsa_ noprint;
          tables trtpn*resp50fl/cmh;
          output out=cocm(keep=N P_CMHGA) cmh;
    run;
```

TIME TO PSA PROGRESSION

Time to PSA progression is assessed by survival analysis. Survival analysis takes into consideration of censored data. Censoring occurs when only partial information about time to event of interest is available [5]. The objectives of survival analysis primarily include estimation of time-to-event, comparison of survival experience among different groups, and evaluation of the relationship between covariates and survival rates.

The general data structure of survival analysis consists of three parts [4]:

1. **Timing variable** T (AVAL = ADT - date of origin + 1): time at event (date of event - date of the origin + 1) or time at last event-free observation (date of censoring - date of the origin + 1), which is a random variable with a probability distribution
2. **Event/Censoring variable** (CNSR): CNSR = 0 if had the event and CNSR = 1 if no event by the analysis cutoff.
3. **Survival function**, the probability of surviving beyond t:

   $S(t) = P(T > t) = 1 - P(T <= t) = 1 - F(t)$, where $F(t)$ is the cdf of the random variable T (the event time for an individual). The difference among different survival models are the choices of distribution for the timing variable T.

According to Prostate Cancer Working Group 2 (PCWG2) guidelines [2], time to PSA progression is defined as the time from randomization to the date of first PSA value demonstrating progression, or to the date of the first confirmed PSA progression. Furthermore, depending **on whether or not the subjects with PSA declines** at the first assessment specified in statistical analysis plan (SAP), PCWG2 defines PSA progression as the record that a >= 25% increase and an absolute increase of >=2 ng/ml **above the nadir** or **above baseline** is documented, which is confirmed by a second consecutive value obtained at least 3 weeks later.

Censoring occurs when no post-baseline PSA values meet the criteria for PSA progression or when patients have no baseline or post-baseline PSA values. For those subjects without confirmed PSA progression at the time of analysis cutoff, they will be right censored (the subject survived at least to the analysis cutoff date) at the date of last PSA assessment before the analysis cutoff date. For those subjects who have no baseline or post-baseline PSA values, date of censoring will be the randomization date.

Figure 2 is a demonstration of event/censoring for time to PSA progression.



Figure 2. Demonstration of Event/Censoring for Time to PSA Progression

Of note, time to progression (TTP) and progression free survival (PFS) endpoints are similar to each other. However, the difference between TTP and PFS is that TTP only considers radiological progressions while PFS includes both radiological progressions and deaths caused by any event [6].

An example of such a summary table is shown below. Time to PSA progression is performed in the intent-to-treat (ITT) population. The table provides a summary of number of PSA progression events and censoring, the estimate of the distribution of duration of time to PSA progression, and the estimate of the follow-up time for all patients. An example of a reporting table for time to PSA Progression is shown below in the next page:

Table 2.0
Time to Prostate-Specific Antigen Progression
(ITT Population)

	Treatment(N=XXX)	Placebo(N=XXX)	Treatment Comparison (Drug vs. Placebo)
Status of PSA Follow-up			
PSA Progression	XXX(XX.X%)	XXX(XX.X%)	
Censored	XXX(XX.X%)	XXX(XX.X%)	
Time to PSA Progression (in Months)			
N	XXX	XXX	
25th Percentile	XX.X	XX.X	
Median (95% CI)	XX.X(XX.X, XX.X)	XX.X(XX.X, XX.X)	
75th Percentile	XX.X	XX.X	
p-value			XXX
Hazard Ratio (95% CI)			XX.X(XX.X, XX.X)
Follow-up Time Based on Reverse Kaplan-Meier Estimates for All Patients (in Months)			
N	XXX	XXX	
25th Percentile	XX.X	XX.X	
Median	XX.X	XX.X	
75th Percentile	XX.X	XX.X	

Generally, the non-parametric estimate Kaplan-Meier (KM) method is used in estimating the distribution of duration of time to PSA progression. Additionally, the median duration of time to PSA progression is estimated by the 50th percentile of Kaplan-Meier estimates along with a 95% confidence interval. The comparison of time to PSA progression between the treatment arm and placebo arm is reported by a p-value based on a log-rank test. The hazard ratio is estimated using a stratified or unstratified cox regression model. Also, the follow-up time can be calculated based on observed follow-up time for censored subjects or can be estimated based on Reverse Kaplan-Meier [5], which reverses the event and censoring cases.

With placebo as the reference group (trt01pn=1 as treatment arm and trt01pn=2 as placebo), hazard ratio can be estimated based on a Cox regression model with treatment as the only covariate, where a hazard ratio < 1 indicates the treatment arm prolongs the time to PSA compared to placebo. A sample dataset for Time to PSA Progression Analysis is shown below:

USUBJID	TRT01PN	PARAMCD	PARAM	ADT	AVAL	CNSR	EVNTDESC
101	1	PDCNCFL	Confirmed PSA Progression	06Mar2019	7.436	0	PSA Progression
102	2	PDCNCFL	Confirmed PSA Progression	29Mar2019	8.213	1	Censored (No PSA Progression)
103	2	PDCNCFL	Confirmed PSA Progression	21Apr2019	32.874	1	Censored (No Baseline or Post-baseline records)

Note: AVAL= (ADT-RANDDT+1)/30.4375 (analysis day in month)

Some useful *SAS*® function for the analysis of Time to PSA progression is attached below:

```
*****************************
  Time to PSA Progression
*****************************;
   ods listing close;
   ods output Lifetest.CensoredSummary=sum(keep=stratum trt01pn total
                                           censored where=(trt01pn in (1 2)));
     proc lifetest data=adttee;
      time aval*cnsr(1);
      strata trt01pn;
     run;
   ods output close;
   ods listing;
***********************
  Log rank p values
***********************;
   ods listing close;
   ods output HomTests=pval(where=(test='Log-Rank'));
     proc lifetest data = adttee;
       time aval*cnsr(1);
       strata trt01pn;
     run;
   ods output close;
   ods listing;
***********************
  Hazard Ratio
***********************;
   ods listing close;
   ods output parameterestimates=unhz(keep=hazardratio hrlowercl
                                      hruppercl);
     proc phreg data = adttee;
       class trt01pn;
       model aval*cnsr(1)=trt01pn/ties=discrete risklimit;
     run;
   ods output close;
*********************************************
  Follow-up Time Based on Reverse Kaplan-Meier
*********************************************;
   proc lifetest data=adttee;
    time aval*cnsr(0);
    strata trt01pn;
   run;
```

(reversing the events and censoring cases)

Notice that when calculating hazard ratio in *SAS*®, one can choose different options in the MODEL statement to specify the methods for handling ties (when two events occur concurrently). The commonly-used methods are TIES=EXACT or TIES=DISCRETE, which require confirmation from statisticians. A heuristic understanding of these two methods is that the Exact method assumes time is a continuous variable and ties occurs due to the imprecise nature of measurement in time, while Discrete method assumes that time is discrete and ties truly occur simultaneously [4].

WATERFALL PLOT

A waterfall plot of best percentage change in PSA from baseline is one of the most popular graphical representations of PSA values. The graph summarizes the maximum percentage reduction in PSA for each individual subject.

Figure 3 is a waterfall plot of best percentage change from baseline in PSA.



Figure 3. Sample Waterfall Plot

DESIGN OF ADAM DATA SET: ADPSA BASED ON THE MOCK-UPS

Based on the mock-up tables and figures, and with the compliance of ADaM ADTTE (Time-to-Event analysis data set) standard data structure, the analysis dataset ADPSA can be designed in the following way. The key variables in ADPSA dataset are USUBJID, TRTP, PARAMCD, PARAM, ADT, AVAL, AVALC, CNSR, EVNTDESC, and CNSDTDSC [7]. For demonstration purposes, in this paper, time to PSA progression is defined as the time from randomization to the date of initial PSA progression.

Table 3 is a sample ADPSA Metadata.

Variable Name	**Variable Label**	**Comment**
NADIR	PSA nadir to-date	Lowest prior value to date starting with first post-baseline value (missing for baseline and earlier).
PARAMCD	Parameter Code	Set to ‘PSA’, ‘PSAEVAL’, ‘PDCNCFL’, ‘P50CNFFL’ accordingly
PARAM	Parameter	PARAMCD=’PSA’: Set to the concatenation of LB.LBTEST with non-missing LB.LBSTRESU in parenthesis. If LB.LBSTRESU is missing, then set to LB.LBTEST PARAMCD=’PSAEVAL’: Set to ‘PSA Evaluable Population’ PARAMCD=’PDCNCFL’: Set to ‘Confirmed PSA Progression’ PARAMCD=’P50CNFFL’: Set to ‘Confirmed Percent Decrease >=50’

AVAL	Analysis Value	PARAMCD='PDCNCFL': (ADT-RANDDT+1)/30.4375
AVALC	Analysis Value (C)	PARAMCD='PSA': LB.LBSTRESC where LB.LBTEST='Prostate Specific Antigen' PARAMCD='PSAEVAL': Create one record for each patient in the ITT population (ITTFL='Y'). If patient has PSA values for both baseline (AVISIT='BASELINE') and postbaseline (AVISIT > 0) then AVALC='Y'. If patient does not have PSA values for either baseline or postbaseline then AVALC='N' PARAMCD='PDCNCFL': Create one record for each patient in the ITT population (ITTFL='Y'). Set AVALC to 'Y' if patient has confirmed PSA progression (defined in EVNTDESC below). If patient has no confirmed PSA progression, set AVALC to 'N' PARAMCD='P50CNFFL': Create one record for each patient in the PSA Evaluable population (AVALC='Y' where PARAMCD='PSAEVAL'). Set AVALC to 'Y' if patient has any postbaseline PSA values with a percent decrease from baseline >=50%, with a consecutive PSA assessment taken >=21 days later also has a decrease from baseline >=50% (confirmation). If patient has no confirmed PSA response, set AVALC to 'N'
ADT	Analysis Date	PARAMCD='PSA': Set to the Numeric date portion of LB.LBDTC PARAMCD='PSAEVAL': Leave null PARAMCD='PDCNCFL': - If subject has confirmed PSA progression (CNSR=0), then set ADT to LB.LBDT of the record showing initial progression(not the record that confirmed progression). - If subject has no baseline or post-baseline PSA values, then set ADT to ADSL.RANDDT - if no PSA progression, then set ADT to LB.LBDT for the last PSA assessment indicating no confirmed PSA progression, or the analysis cutoff date, whichever occurs first. PARAMCD='P50CNFFL': - If response (AVALC='Y'), set ADT to ADT of the record showing initial response (not the record that confirmed response). - If no response (AVALC='N') then leave null
EVNTDESC	Event or Censoring Description	EVNTDESC = 'PSA Progression' if: 1. **For the patients who have PSA decrease from baseline at Week 1 visit window (specified in SAP)**, if (a PSA value with an increase above **nadir** >=25% and an increase above **nadir** >=2ng/ml), that is confirmed with a consecutive value meet such criteria taken >=3 weeks later, then set EVNTDESC = 'PSA Progression'. 2. **For the patients who have no post-baseline PSA value at Week 1 but have PSA value after Week 1 visit window,** with (an increase above **nadir** >=25% and an increase above **nadir** >=2ng/ml), that is confirmed with a consecutive value meet such

		criteria taken >=3 weeks later, then set EVNTDESC = 'PSA Progression'. 3. **For the patients whose PSA assessment >= baseline at Week 1**, if (a PSA value with an increase above **base** >=25% and an increase above **base** >=2ng/ml), that is confirmed with a consecutive value meet such criteria taken >=3 weeks later, then set EVNTDESC = 'PSA Progression'. If subject has no baseline or post-baseline PSA values, or no post-baseline PSA value meet the PSA progression criteria, then set to 'Censored'
CNSR	Censor	Derived based on EVNTDESC: Set to 1 if EVNTDESC = 'Censored' Set to 0 if EVNTDESC = 'PSA Progression'

Table 3. ADPSA Metadata

Following the metadata structure from above, one example subject data in ADPSA dataset are presented below (the codes for generating such a dataset are in the appendix):

USUBJID	PARAMCD	PARAM	PARAMTYP	AVISIT	AVISITN	ADT	BASE	AVAL	AVALC	CHBLP	NADIR	REFAVAL	CHGREF	PCHGREF	CNSR	EVNTDESC
101	PSA			SCREENING	-1	2019-06-23		3.31								
101	PSA			BASELINE	0	2019-07-05	4.23	4.23								
101	PSA			WEEK 1	1	2019-07-22	4.23	2.31		-45.39	4.23	4.23	-1.92	-45.39		
101	PSA			WEEK 4	2	2019-08-02	4.23	7.13		68.56	2.31	2.31	4.82	208.66		
101	PSA			WEEK 9	3	2019-09-06	4.23	18.21		330.5	2.31	2.31	15.9	688.31		
101	PSA			WEEK 15	4	2019-11-04	4.23	13.35		215.6	2.31	2.31	11.04	477.92		
101	PSA			WEEK 19	5	2019-12-02	4.23	16.52		290.54	2.31	2.31	14.21	615.15		
101	PSAEVAL	PSA Evaluable Population	DERIVED						Y							
101	PDCNCFL	Confirmed PSA Progression	DERIVED			2019-08-02		0.95277	Y						0	PSA Progression
101	P50CNFFL	Confirmed Percent Decrease >=50	DERIVED						N							

ADPSA DATASET CALCULATION CONSIDERATION

1. Concept of NADIR: Nadir is defined as the lowest value prior to current assessment and is derived at each time point. Baseline records (records with ABLFL='Y') can be used as nadir value, but no records prior to baseline value should be considered.

2. When deriving Confirmed PSA Progression, it will simplify the codes by deriving a reference value (REFAVAL) as follows:

The values of REFAVAL are missing for all with ADY prior to the week 1 visit window.

Starting with the beginning of the week 1 visit window, the reference value is calculated based on the earliest PSA value in the week 1 window:

- If the earliest PSA value at Week 1 >= BASE, then the REFAVAL= BASE

- If the earliest PSA value at Week 1 < BASE, then the REFAVAL= NADIR

- If there are no PSA values in the week 1 window, then the REFAVAL= NADIR

If defining CHGREF as AVAL-REFAVAL and PCHGREF as 100*(CHGREF/REFAVAL), then for non-missing postbaseline PSA values (after week 1) with PCHGREF >= 25 and CHGREF >= 2, it is a single PSA progression. This single PSA progression requires to be confirmed by a consecutive value at least 3 weeks later.

APPENDIX

Sample code for deriving NADIR and PARAM= 'Confirmed PSA Progression':

```
data psa;
  format RANDDT mmddyy10.;
  input USUBJID $ 1-3 AVISIT $ 5-13 AVISITN LBDTC $ 19-28 PSAAVAL;
  retain BASE;
  RANDDT='05JUL2019'd;
  if avisitn=0 then base=psaaval;
  if avisitn > 0 then postbfl=1; else postbfl=0;
  ADT=input(LBDTC,yymmdd10.);
  datalines;
  101 SCREENING  -1 2019-06-23 3.31
  101 BASELINE   0  2019-07-05 4.23
  101 WEEK 1     1  2019-07-22 2.31
  101 WEEK 4     2  2019-08-02 7.13
  101 WEEK 9     3  2019-09-06 18.21
  101 WEEK 15    4  2019-11-04 13.35
  101 WEEK 19    5  2019-12-02 16.52;
  run;

* nadir ;
proc sort data=psa;
  by usubjid postbfl;
run;
data psa;
  set psa;
  by usubjid postbfl;
    retain NADIR;
    if first.usubjid then nadir=.;
    lagpsa=lag(psaaval);
    if postbfl then do;
          if first.postbfl then nadir=base;
          else nadir=min(nadir,lagpsa);
    end;
    if postbfl and psaaval ne . then do;
          chg_nadir=psaaval-nadir;
          if nadir ne 0 then pchg_nadir=chg_nadir/nadir*100;
    end;
run;

* week 1 value for refaval variable;
proc sort data=psa out=wk1val;
```

```
    by usubjid avisitn lbdtc;
    where avisitn=1;
run;
data wk1val;
  set wk1val;
  by usubjid avisitn lbdtc;
    if avisitn = 1 and first.avisitn;
    if psaaval < base then nadflag = 1;
    else if psaaval >= base then nadflag = 2;
  keep usubjid nadflag;
run;
data final_;
  merge psa(in=a) wk1val(in=b);
  by usubjid;
    if a and not b then nadflag = 3;
    if avisitn in (-1,0) then nadir = .;
    if avisitn >= 1 then do;
          if nadflag in (1,3) then refaval = nadir;
          else if nadflag = 2 then refaval = base;
          chgref = psaaval - refaval;
          pchgref = 100*(chgref/refaval);
    end;
    if chgref >= 2 and pchgref >= 25 and avisitn >= 1 then psaprogfl = 'Y';
run;

/*PDCNCFL: Single PSA progression has happened twice consecutively over a
>= 21 day span. In other words, an initial single PSA progression must be
confirmed 3 weeks later*/
proc sort data=final_;by usubjid adt;run;
data PDCNCFL;
  set final_;
  by usubjid lbdtc avisitn;
    lagadt=lag(adt);
    lagpsaprogfl=lag(psaprogfl);
    llagpsaprogfl =lag(lagpsaprogfl);
    if first.usubjid then do;
          lagadt='';
          lagpsaprogfl='';
          llagpsaprogfl='';
    end;

    diffadt=(adt-lagadt)+1;
    if psaprogfl='Y' and diffadt>21 and lagpsaprogfl='Y' then PDCNCFL='Y';
    if psaprogfl='Y' and lagpsaprogfl='Y' and llagpsaprogfl='Y' then
    PDCNCFL='Y';

    if psaprogfl='Y';
run;
data adpsa_;
  format adt date9.;
  set PDCNCFL;;
  by usubjid lbdtc avisitn;
    if first.usubjid;
    ADT=input(lbdtc,yymmdd10.);
    AVAL=(ADT-RANDDT+1)/30.4375;
    EVNTDESC='Confirmed PSA Progression';
    CNSR = 0;
```

```
    keep usubjid lbdtc adt randdt aval evntdesc cnsr;
  run;
  data adpsa;
    retain USUBJID LBDTC ADT RANDDT AVAL EVNTDESC CNSR;
    set adpsa_;
  run;
```

The codes for deriving for PARAM='Confirmed Percent Decrease >=50' is similar. Derive a flag variable for single percent decrease >=50 and then check for confirmation.

CONCLUSION

While programming in Oncology is specialized, having a study specific guideline at a programmers' disposal can streamline the process and make it straightforward. This paper provides a guideline for the major PSA efficacy analyses from data collection to mock-up tables, and summary table reporting. We discuss different scenarios when defining PSA response and different event/censoring cases when calculating time to PSA progression. Additionally, we share the necessary statistical knowledge and programming techniques that are required in table programming and considerations for dataset derivation.

REFERENCES

[1] Cancer Facts & Figures 2019. (n.d.). Retrieved November 20, 2019, from https://www.cancer.org/content/dam/cancer-org/research/cancer-facts-and-statistics/annual-cancer-facts-and-figures/2019/cancer-facts-and-figures-2019.pdf

[2] Scher, H. I., Halabi, S., Tannock, I., Morris, M., Sternberg, C. N., & Carducci, M. A. (2008, March 1). Design and End Points of Clinical Trials for Patients With ... Retrieved from https://imaging.cancer.gov/clinical_trials/docs/Prostate Cancer Working Group 2 - bone response criteria.pdf

[3] Clopper, C. J., & Pearson, E. S. (1934). The Use of Confidence or Fiducial Limits Illustrated in the Case of the Binomial. *Biometrika,26*(4), 404. doi:10.2307/2331986

[4] ALLISON, P. D. (2018). SURVIVAL ANALYSIS USING SAS: A practical guide, second edition. S.l.: SAS INSTITUTE.

[5] Prinja, S., Gupta, N., & Verma, R. (2010). Censoring in clinical trials: Review of survival analysis techniques. *Indian Journal of Community Medicine,35*(2), 217. doi:10.4103/0970-0218.66859

[6] Efficacy Endpoints in Oncology Clinical Trials. (n.d.). Retrieved from https://www.biooncology.com/clinical-trials/efficacy-endpoints.html

[7] The ADaM Basic Data Structure for Time-to-Event Analyses. (n.d.). Retrieved November 20, 2019, from https://www.cdisc.org/system/files/all/standard_category/application/pdf/adam_tte_final_v1.pdf.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Joy Zeng
Pfizer Inc
jingying.zeng@pfizer.com

Varaprasad Ilapogu
Ephicacy Consultancy Group
prasad.ilapogu@pfizer.com

Xinping Cindy Wu
Pfizer Inc
Xinping.Wu@pfizer.com